Title

newey postestimation — Postestimation tools for newey

Postestimation commands predict margins Remarks and examples
Also see

Postestimation commands

The following postestimation commands are available after `newey`:

Command	Description
`contrast`	contrasts and ANOVA-style joint tests of estimates
`estat summarize`	summary statistics for the estimation sample
`estat vce`	variance–covariance matrix of the estimators (VCE)
`estimates`	cataloging estimation results
`etable`	table of estimation results
`forecast`	dynamic forecasts and simulations
`lincom`	point estimates, standard errors, testing, and inference for linear combinations of coefficients
`linktest`	link test for model specification
`margins`	marginal means, predictive margins, marginal effects, and average marginal effects
`marginsplot`	graph the results from margins (profile plots, interaction plots, etc.)
`nlcom`	point estimates, standard errors, testing, and inference for nonlinear combinations of coefficients
`predict`	predictions and their SEs, residuals, etc.
`predictnl`	point estimates, standard errors, testing, and inference for generalized predictions
`pwcompare`	pairwise comparisons of estimates
`test`	Wald tests of simple and composite linear hypotheses
`testnl`	Wald tests of nonlinear hypotheses

predict

Description for predict

`predict` creates a new variable containing predictions such as linear predictions and residuals.

Menu for predict

Statistics > Postestimation

Syntax for predict

`predict` [*type*] *newvar* [*if*] [*in*] [, *statistic*]

statistic	Description
Main	
`xb`	linear prediction; the default
`stdp`	standard error of the linear prediction
`residuals`	residuals

These statistics are available both in and out of sample; type `predict ... if e(sample) ...` if wanted only for the estimation sample.

Options for predict

Main

`xb`, the default, calculates the linear prediction.

`stdp` calculates the standard error of the linear prediction.

`residuals` calculates the residuals.

margins

Description for margins

margins estimates margins of response for linear predictions.

Menu for margins

Statistics > Postestimation

Syntax for margins

margins [*marginlist*] [, *options*]

margins [*marginlist*] , predict(*statistic* ...) [*options*]

statistic	Description
xb	linear prediction; the default
stdp	not allowed with margins
residuals	not allowed with margins

Statistics not allowed with margins are functions of stochastic quantities other than e(b).

For the full syntax, see [R] **margins**.

Remarks and examples

stata.com

▷ Example 1

We use the test command after newey to illustrate the importance of accounting for the presence of serial correlation in the error term. The dataset contains daily stock returns of three car manufacturers from January 2, 2003, to December 31, 2010, in the variables toyota, nissan, and honda.

We fit a model for the Nissan stock returns on the Honda and Toyota stock returns, and we use estat bgodfrey to test for serial correlation of order one:

```
. use https://www.stata-press.com/data/r18/stocks
(Data from Yahoo! Finance)

. regress nissan honda toyota
  (output omitted)

. estat bgodfrey
Breusch-Godfrey LM test for autocorrelation
```

lags(p)	chi2	df	Prob > chi2
1	6.415	1	0.0113

```
                H0: no serial correlation
```

The result implies that the error term is serially correlated; therefore, we should rather fit the model with `newey`. But let's use the outcome from `regress` to conduct a test for the statistical significance of a particular linear combination of the two coefficients in the regression:

```
. test 1.15*honda+toyota = 1
 ( 1)  1.15*honda + toyota = 1
       F(  1,  2012) =    5.52
            Prob > F =    0.0189
```

We reject the null hypothesis that the linear combination is valid. Let's see if the conclusion remains the same when we fit the model with `newey`, obtaining the Newey–West standard errors for the OLS coefficient estimates.

```
. newey nissan honda toyota,lag(1)
  (output omitted)
. test 1.15*honda+toyota = 1
 ( 1)  1.15*honda + toyota = 1
       F(  1,  2012) =    2.57
            Prob > F =    0.1088
```

The conclusion would be the opposite, which illustrates the importance of using the proper estimator for the standard errors.

◁

▷ Example 2

We want to produce forecasts based on dynamic regressions for each of the three stocks. We will treat the stock returns for `toyota` as a leading indicator for the two other stocks. We also check for autocorrelation with the Breusch–Godfrey test.

```
. use https://www.stata-press.com/data/r18/stocks
(Data from Yahoo! Finance)
. regress toyota l(1/2).toyota
  (output omitted)
. estat bgodfrey
Breusch-Godfrey LM test for autocorrelation
```

lags(*p*)	chi2	df	Prob > chi2
1	4.373	1	0.0365

```
H0: no serial correlation
. regress nissan l(1/2).nissan l.toyota
  (output omitted)
. estat bgodfrey
Breusch-Godfrey LM test for autocorrelation
```

lags(*p*)	chi2	df	Prob > chi2
1	0.099	1	0.7536

```
H0: no serial correlation
```

```
. regress honda l(1/2).honda l.toyota
  (output omitted)
. estat bgodfrey
Breusch-Godfrey LM test for autocorrelation
```

lags(p)	chi2	df	Prob > chi2
1	0.923	1	0.3367

```
                        H0: no serial correlation
```

The first result indicates that we should consider using `newey` to fit the model for `toyota`. The point forecasts would not be actually affected because `newey` produces the same OLS coefficient estimates reported by `regress`. However, if we were interested in obtaining measures of uncertainty surrounding the point forecasts, we should then use the results from `newey` for that first equation.

Let's illustrate the use of `forecast` with `newey` for the first equation and `regress` for the two other equations. We first declare the forecast model:

```
. forecast create stocksmodel
  Forecast model stocksmodel started.
```

Then we refit the equations and add them to the forecast model:

```
. quietly newey toyota l(1/2).toyota, lag(1)
. estimates store eq_toyota
. forecast estimates eq_toyota
  Added estimation results from newey.
  Forecast model stocksmodel now contains 1 endogenous variable.
. quietly regress nissan l(1/2).nissan l.toyota
. estimates store eq_nissan
. forecast estimates eq_nissan
  Added estimation results from regress.
  Forecast model stocksmodel now contains 2 endogenous variables.
. quietly regress honda l(1/2).honda l.toyota
. estimates store eq_honda
. forecast estimates eq_honda
  Added estimation results from regress.
  Forecast model stocksmodel now contains 3 endogenous variables.
```

We use `tsappend` to add the number of periods for the forecast, and then we obtain the predicted values with `forecast solve`:

```
. tsappend, add(7)
. forecast solve, prefix(stk_)
Computing dynamic forecasts for model stocksmodel.
───────────────────────────────────────────────────
Starting period:  2016
Ending period:    2022
Forecast prefix:  stk_

2016:  ............
2017:  ...........
2018:  ...........
2019:  ..........
2020:  .........
2021:  ........
2022:  ........

Forecast 3 variables spanning 7 periods.
─────────────────────────────────────────
```

The graph below shows several interesting results. First, the stock returns of the competitor (`toyota`) does not seem to be a leading indicator for the stock returns of the two other companies (otherwise, the patterns for the movements in `nissan` and `honda` would be following the recent past movements in `toyota`). You can actually fit the models above for `nissan` and `honda` to confirm that the coefficient estimate for the first lag of `toyota` is not significant in any of the two equations. Second, immediately after the second forecasted period, there is basically no variation in the predictions, which indicates the very short-run predicting influence of past history on the forecasts of the three stock returns.



◁

Also see

[TS] **newey** — Regression with Newey–West standard errors

[U] 20 Estimation and postestimation commands

For suggested citations, see the FAQ on citing Stata documentation.